```
libname raw '/home/u58485303/raw';
libname output '/home/u58485303/output';

/* demog raw data */
/* created variables STUDYID DOMAIN SUBJID SITEID USUBJID BRTHDTC SEX RACE COUNTRY */
data dm1;
set raw.demog (rename=(SEX= SEX1 RACE=RACE1));

STUDYID= 'ACP-103-020';
DOMAIN= 'DM';
SUBJID= strip(substr(SUBJECT, 4,3));
SITEID= strip(substr(SITE, 5,3));
USUBJID= compress(STUDYID || '-' || SITEID|| '-' || SUBJID);

BRTHDTC= put(input(BRTHDT,??DATE11.),??IS8601DA.);

if SEX1 eq 1 then SEX="M";
else if SEX1 eq 2 then SEX="F";

if RACE1 eq 1 then RACE="WHITE";
else if RACE1 eq  2 then RACE="BLACK OR AFRICAN AMERICAN";
else if RACE1 eq 3 then RACE="ASIAN";
else if RACE1 in (4 99) then RACE="OTHER";

COUNTRY= "USA";

keep STUDYID DOMAIN SUBJID SITEID USUBJID BRTHDTC SEX RACE COUNTRY RACEOTH;
run;
```

```
/* inco raw data */
/* created variable RFICDTC */
data dm2;
set raw.inco;

STUDYID= 'ACP-103-020';
DOMAIN= 'DM';
SUBJID= strip(substr(SUBJECT, 4,3));
SITEID= strip(substr(SITE, 5,3));
USUBJID= compress(STUDYID || '-' || SITEID|| '-' || SUBJID);

RFICDTC= ICFDATE;
keep USUBJID RFICDTC;
run;

/* drug raw data */
/* created variables RFSTDTC RFXSTDTC ACTARM ACTARMCD */

proc sort data=raw.drug out=drug (keep= SUBJECT SITE DOSEDTN TRTMNT);
by SUBJECT DOSEDTN TRTMNT;
run;

data dm3;
set raw.drug;
WHERE DOSEDTN ne . AND TRTMNT ne "";

STUDYID= 'ACP-103-020';
DOMAIN= 'DM';
SUBJID= strip(substr(SUBJECT, 4,3));
```

```
SITEID= strip(substr(SITE, 5,3));
USUBJID= compress(STUDYID || '-' || SITEID|| '-' || SUBJID);

by SUBJECT DOSEDTN;
if first.SUBJECT;

RFSTDTC= put(DOSEDTN, ??IS8601DA.);
RFXSTDTC= RFSTDTC;

ACTARM= UPCASE(TRTMNT);
if ACTARM='TRIAL DRUG A' then ACTARMCD= 'TDRUGA';
if ACTARM='TRIAL DRUG B' then ACTARMCD= 'TDRUGB';

keep USUBJID RFSTDTC RFXSTDTC ACTARM ACTARMCD;
run;

/* created variables RFENDTC RFXENDTC */

data dm4;
set raw.drug;
WHERE DOSEDTN ne . AND TRTMNT ne "";

STUDYID= 'ACP-103-020';
DOMAIN= 'DM';
SUBJID= strip(substr(SUBJECT, 4,3));
SITEID= strip(substr(SITE, 5,3));
USUBJID= compress(STUDYID || '-' || SITEID|| '-' || SUBJID);

by SUBJECT DOSEDTN;
```

```
if last.SUBJECT;

RFENDTC= put(DOSEDTN, ??IS8601DA.);
RFXENDTC= RFENDTC;

keep USUBJID RFENDTC RFXENDTC;
run;

/* visit raw data */
/* created variable RFPENDTC */

data visit1;
set raw.visit;
VISITDT1=put(input(VISITDT, ??DATE11.), ??IS8601DA.);
run;

proc sort data=visit1 out= visit (keep= SUBJECT SITE VISITDT1);
by SUBJECT VISITDT1;
run;

data dm5;
set visit;
WHERE VISITDT1 ne "";

STUDYID= 'ACP-103-020';
DOMAIN= 'DM';
SUBJID= strip(substr(SUBJECT, 4,3));
SITEID= strip(substr(SITE, 5,3));
USUBJID= compress(STUDYID || '-' || SITEID|| '-' || SUBJID);
```

```
by SUBJECT VISITDT1;
if last.SUBJECT;

RFPENDTC= VISITDT1;

keep USUBJID RFPENDTC;
run;

/* random raw data */
/* created variables ARM ARMCD */

proc import datafile= '/home/u58485303/raw/random.xlsx' out= raw.random
dbms= XLSX replace;
quit;

data dm6;
set raw.random;

STUDYID= 'ACP-103-020';
DOMAIN= 'DM';
SUBJID= strip(substr(SUBJECT, 4,3));
SITEID= strip(substr(SUBJECT, 1,3));
USUBJID= compress(STUDYID || '-' || SITEID|| '-' || SUBJID);

ARM= RANDTRT;
if ARM= 'TRIAL DRUG A' then ARMCD= 'TDRUGA';
else if ARM= 'TRIAL DRUG B' then ARMCD= 'TDRUGB';
keep USUBJID ARM ARMCD;
```

```
run;

/* ds_raw raw data */
/* filtered and retained only screen failures  */

data dm7;
set raw.ds_raw;
WHERE SCRN= 'Fail';

STUDYID= 'ACP-103-020';
DOMAIN= 'DM';
SUBJID= strip(substr(SUBJECT, 4,3));
SITEID= strip(substr(SITE, 5,3));
USUBJID= compress(STUDYID || '-' || SITEID|| '-' || SUBJID);

keep USUBJID;
run;

/* sorted all temp dm datasets created */

proc sort data=dm1;
by USUBJID;
run;

proc sort data=dm2;
by USUBJID;
run;

proc sort data=dm3;
```

```
by USUBJID;
run;

proc sort data=dm4;
by USUBJID;
run;

proc sort data=dm5;
by USUBJID;
run;

proc sort data=dm6;
by USUBJID;
run;

proc sort data=dm7;
by USUBJID;
run;

/*merged all temp dm datasets. Created ARM ARMCD ACTARM ACTARMCD variables as required
for screen failure, not treated and not assigned. Creatd variables AGE AGEU */

data all;
length ARM ARMCD ACTARM ACTARMCD $15.;
merge dm1 (in=A) dm2 dm3 dm4 dm5 dm6 dm7 (in=G);
by USUBJID;
if A;

if A and G then do;
```

```
        ARM= 'Screen Failure';
        ARMCD ='SCRNFAIL';
        ACTARM= 'Screen Failure';
        ACTARMCD ='SCRNFAIL';
        end;

else if ACTARM= '' then do;
        ACTARM= 'Not Treated';
        ACTARMCD ='NOTTRT';
        end;

else if ARM='' then do;
        ARM= 'Not Assigned';
        ARMCD ='NOTASSIG';
        end;

AGE= ROUND((INPUT(RFSTDTC,??IS8601DA.)- INPUT(BRTHDTC,??IS8601DA.)+1)/365.25);
AGEU= 'YEARS';
run;

/* assigned attributes and created final dm dataset */

data output.dm (label= 'Demographics');
retain STUDYID DOMAIN USUBJID SUBJID RFSTDTC RFENDTC RFXSTDTC RFXENDTC RFICDTC
RFPENDTC SITEID BRTHDTC AGE AGEU SEX RACE ARMCD     ARM     ACTARMCD ACTARM
        COUNTRY;
attrib STUDYID label= 'Study Identifier' length= $20
                DOMAIN label= 'Domain Abbreviation' length= $2
                USUBJID label= 'Unique Subject Identifier' length= $40
```

```
                SUBJID label= 'Subject Identifier for the Study' length= $10
                RFSTDTC label= 'Subject Reference Start Date/Time' length= $20
                RFENDTC label= 'Subject Reference End Date/Time' length= $20
                RFXSTDTC label= 'Date/Time of First Study Treatment' length= $20
                RFXENDTC label= 'Date/Time of Last Study Treatment' length= $20
                RFICDTC label= 'Date/Time of Informed Consent' length= $20
                RFPENDTC label= 'Date/Time of End of Participation' length= $20
                SITEID label= 'Study Site Identifier' length= $3
                BRTHDTC label= 'Date/Time of Birth' length= $20
                AGE label= 'Age' length= 8
                AGEU label= 'Age Units' length= $5
                SEX label= 'Sex' length= $1
                RACE label= 'Race' length= $30
                ARMCD label= 'Planned Arm Code' length= $30
                ARM label= 'Description of Planned Arm' length= $100
                ACTARMCD label= 'Actual Arm Code' length= $30
                ACTARM label= 'Description of Actual Arm' length= $100
                COUNTRY label= 'Country' length= $3;
set all;

keep STUDYID DOMAIN          USUBJID          SUBJID RFSTDTC RFENDTC RFXSTDTC
        RFXENDTC RFICDTC RFPENDTC SITEID BRTHDTC AGE AGEU SEX RACE ARMCD ARM
ACTARMCD ACTARM COUNTRY;
run;

/* proc compare base=output.dm1 comp=output.dm listall;
ods listing file= '/home/u58485303/output/dm.lst';
run; */
```

```
/* created suppdm */

data output.suppdm (label= 'Supplemental Qualifiers for DM');
retain STUDYID RDOMAIN USUBJID IDVAR IDVARVAL QNAM QLABEL QVAL QORIG QEVAL;
attrib STUDYID label= 'Study Identifier' length= $20
                RDOMAIN label= 'Related Domain Abbreviation' length= $2
                USUBJID label= 'Unique Subject Identifier' length= $40
                IDVAR label= 'Identifying Variable' length= $5
                IDVARVAL label= 'Identifying Variable Value' length= $3
                QNAM label= 'Qualifier Variable Name' length= $8
                QLABEL label= 'Qualifier Variable Label' length= $40
                QVAL label= 'Data Value' length= $20
                QORIG label= 'Origin' length= $3
                QEVAL label= 'Evaluator' length= $1;
set all;

if RACEOTH ne '' then do;
RDOMAIN= 'DM';
IDVAR= '';
IDVARVAL= '';
QNAM= 'RACEOTH';
QLABEL="Race Other Specify";
QVAL= UPCASE(RACEOTH);
QORIG= 'CRF';
QEVAL= '';
output;
end;
```

```
keep STUDYID RDOMAIN USUBJID IDVAR IDVARVAL    QNAM QLABELQVAL QORIG QEVAL;
run;

/* proc compare base=output.suppdm1 comp=output.suppdm listall;
ods listing file= '/home/u58485303/output/suppdm.lst';
run; */
```